```
?
```
(1)

In this Maple file, we compute the Casimir coefficients of the Lax matrix L associated to the Painlevé 5 equation in relation with the spectral curve

```
> restart:
  P1:=x-> P011/x+P121/(x-1)^2+P111/(x-1);
  P2:=x-> P022/x^2+P012/x+P142/(x-1)^4+P132/(x-1)^3+P122/(x-1)^2+
  P112/(x-1);
  SpectralCurve:=unapply( y^2-P1(x)*y+P2(x),y);
```

$$P1 := x \to \frac{P011}{x} + \frac{P121}{(x-1)^2} + \frac{P111}{x-1} \tag{2}$$

$$P2 := x \to \frac{P022}{x^2} + \frac{P012}{x} + \frac{P142}{(x-1)^4} + \frac{P132}{(x-1)^3} + \frac{P122}{(x-1)^2} + \frac{P112}{x-1}$$

$$SpectralCurve := y \to y^2 - \left(\frac{P011}{x} + \frac{P121}{(x-1)^2} + \frac{P111}{x-1} \right) y + \frac{P022}{x^2} + \frac{P012}{x} + \frac{P142}{(x-1)^4} + \frac{P132}{(x-1)^3} + \frac{P122}{(x-1)^2} + \frac{P112}{x-1}$$

```
> DiaginftySheet1:=-tinfty10/x+Unknown/x^2;
  DiaginftySheet2:=-tinfty20/x+Unknown2/x^2;
  Diag0Sheet1:=t010/x+Unknown3;
  Diag0Sheet2:=t020/x+Unknown4;
  Diag1Sheet1:=t111/(x-1)^2+t110/(x-1)+Unknown5;
  Diag1Sheet2:=t121/(x-1)^2+t120/(x-1)+Unknown6;
```

$$DiaginftySheet1 := -\frac{tinfty10}{x} + \frac{Unknown}{x^2} \tag{3}$$

$$DiaginftySheet2 := -\frac{tinfty20}{x} + \frac{Unknown2}{x^2}$$

$$Diag0Sheet1 := \frac{t010}{x} + Unknown3$$

$$Diag0Sheet2 := \frac{t020}{x} + Unknown4$$

$$Diag1Sheet1 := \frac{t111}{(x-1)^2} + \frac{t110}{x-1} + Unknown5$$

$$Diag1Sheet2 := \frac{t121}{(x-1)^2} + \frac{t120}{x-1} + Unknown6$$

Expression of P_1 in terms of the diagonal expansion in both sheets

```
> series(DiaginftySheet1+DiaginftySheet2-P1(x),x=infinity);
  series(Diag0Sheet1+Diag0Sheet2-P1(x),x=0,10);
  series(Diag1Sheet1+Diag1Sheet2-P1(x),x=1,10);
  P011:=t010+t020;
  P121:=t111+t121;
  P111:=t110+t120;
  CoherenceEquation1:=tinfty10+tinfty20+P011+P111;
```

$$\frac{-tinfty10 - tinfty20 - P011 - P111}{x} + \frac{Unknown + Unknown2 - P121 - P111}{x^2} + \frac{-2\,P121 - P111}{x^3} + \frac{-3\,P121 - P111}{x^4} + \frac{-4\,P121 - P111}{x^5} + O\left(\frac{1}{x^6}\right) \quad \textbf{(4)}$$

$$\frac{t010 + t020 - P011}{x} + Unknown3 + Unknown4 - P121 + P111 + (-2\,P121 + P111)\,x + (-3\,P121 + P111)\,x^2 + (-4\,P121 + P111)\,x^3 + (-5\,P121 + P111)\,x^4 + (-6\,P121 + P111)\,x^5 + (-7\,P121 + P111)\,x^6 + (-8\,P121 + P111)\,x^7 + (-9\,P121 + P111)\,x^8 + (-10\,P121 + P111)\,x^9 + O(x^{10})$$

$$\frac{t111 + t121 - P121}{(x-1)^2} + \frac{t110 + t120 - P111}{x-1} + Unknown5 + Unknown6 - P011 + P011\,(x-1) - P011\,(x-1)^2 + P011\,(x-1)^3 - P011\,(x-1)^4 + P011\,(x-1)^5 - P011\,(x-1)^6 + P011\,(x-1)^7 - P011\,(x-1)^8 + P011\,(x-1)^9 + O((x-1)^{10})$$

$$P011 := t010 + t020$$
$$P121 := t111 + t121$$
$$P111 := t110 + t120$$
$$CoherenceEquation1 := tinfty10 + tinfty20 + t010 + t020 + t110 + t120$$

Study at 0

```
> factor(series(SpectralCurve(Diag0Sheet1),x=0)):
  factor(series(SpectralCurve(Diag0Sheet2),x=0)):
  EQ01:=residue(x^3*SpectralCurve(Diag0Sheet1),x=0);
  EQ02:=residue(x^3*SpectralCurve(Diag0Sheet2),x=0);
  EQ03:=residue(x^2*SpectralCurve(Diag0Sheet1),x=0);
  EQ04:=residue(x^2*SpectralCurve(Diag0Sheet2),x=0);
  EQ05:=residue(x*SpectralCurve(Diag0Sheet1),x=0);
  EQ06:=residue(x*SpectralCurve(Diag0Sheet2),x=0);
  EQ07:=residue(x^0*SpectralCurve(Diag0Sheet1),x=0);
  EQ08:=residue(x^0*SpectralCurve(Diag0Sheet2),x=0);
```

$$EQ01 := 0$$
$$EQ02 := 0$$
$$EQ03 := 0$$
$$EQ04 := 0$$
$$EQ05 := -t010\,t020 + P022$$
$$EQ06 := -t010\,t020 + P022$$

$$EQ07 := t010\,t110 - t010\,t111 + t010\,t120 - t010\,t121 + t010\,Unknown3 - t020\,Unknown3 + P012$$

$$EQ08 := -t010\,Unknown4 + t020\,t110 - t020\,t111 + t020\,t120 - t020\,t121 + t020\,Unknown4 + P012$$

```
> P022:=solve(EQ05,P022);
  simplify(EQ05);
  simplify(EQ06);
  simplify(EQ07);
```

```
simplify(EQ08);
```

$$P022 := t010\, t020 \tag{6}$$
$$0$$
$$0$$
$$(t110 - t111 + t120 - t121 + Unknown3)\, t010 - Unknown3\, t020 + P012$$
$$(t110 - t111 + t120 - t121 + Unknown4)\, t020 - Unknown4\, t010 + P012$$

Study at infinity

```
> series(SpectralCurve(DiaginftySheet1),x=infinity):
  series(SpectralCurve(DiaginftySheet2),x=infinity):
  EQinftDiag1Sheet1:=residue(x^(-2)*SpectralCurve(DiaginftySheet1),
  x=infinity);
  EQinfty2:=residue(x^(-2)*SpectralCurve(DiaginftySheet2),x=
  infinity);
  EQinfty3:=residue(x^(-1)*SpectralCurve(DiaginftySheet1),x=
  infinity);
  EQinfty4:=residue(x^(-1)*SpectralCurve(DiaginftySheet2),x=
  infinity);
  EQinfty5:=residue(x^(0)*SpectralCurve(DiaginftySheet1),x=
  infinity);
  EQinfty6:=residue(x^(0)*SpectralCurve(DiaginftySheet2),x=
  infinity);
  EQinfty7:=residue(x^(1)*SpectralCurve(DiaginftySheet1),x=
  infinity);
  EQinfty8:=residue(x^(1)*SpectralCurve(DiaginftySheet2),x=
  infinity);
```

$$EQinftDiag1Sheet1 := 0 \tag{7}$$
$$EQinfty2 := 0$$
$$EQinfty3 := 0$$
$$EQinfty4 := 0$$
$$EQinfty5 := -P012 - P112$$
$$EQinfty6 := -P012 - P112$$
$$EQinfty7 := -tinfty10^2 - (t010 + t020 + t110 + t120)\, tinfty10 - t010\, t020 - P122 - P112$$
$$EQinfty8 := -tinfty20^2 - (t010 + t020 + t110 + t120)\, tinfty20 - t010\, t020 - P122 - P112$$

```
> CoherenceEquation2:=-EQinfty5;
  CoherenceEquation3:=-EQinfty7;
  CoherenceEquation4:=-EQinfty8;
  simplify(CoherenceEquation3-CoherenceEquation1*tinfty10);
  CoherenceEquation5:=simplify(CoherenceEquation4-
  CoherenceEquation1*tinfty20);
```

$$CoherenceEquation2 := P012 + P112 \tag{8}$$
$$CoherenceEquation3 := tinfty10^2 + (t010 + t020 + t110 + t120)\, tinfty10 + t010\, t020 + P122 + P112$$
$$CoherenceEquation4 := tinfty20^2 + (t010 + t020 + t110 + t120)\, tinfty20 + t010\, t020 + P122 + P112$$

$$t010\,t020 - tinfty10\,tinfty20 + P112 + P122$$

$$CoherenceEquation5 := t010\,t020 - tinfty10\,tinfty20 + P112 + P122$$

```
> CoherenceEquation1;
  CoherenceEquation2;
  CoherenceEquation5;
```

$$tinfty10 + tinfty20 + t010 + t020 + t110 + t120 \tag{9}$$
$$P012 + P112$$
$$t010\,t020 - tinfty10\,tinfty20 + P112 + P122$$

Study at 1

```
> series(SpectralCurve(Diag1Sheet1),x=1):
  series(SpectralCurve(Diag1Sheet2),x=1):
  EQ19:=residue((x-1)^(3)*SpectralCurve(Diag1Sheet1),x=1);
  EQ20:=residue((x-1)^(3)*SpectralCurve(Diag1Sheet2),x=1);
  EQ21:=residue((x-1)^(4)*SpectralCurve(Diag1Sheet1),x=1);
  EQ22:=residue((x-1)^(4)*SpectralCurve(Diag1Sheet2),x=1);
  EQ11:=residue((x-1)^(2)*SpectralCurve(Diag1Sheet1),x=1);
  EQ12:=residue((x-1)^(2)*SpectralCurve(Diag1Sheet2),x=1);
  EQ13:=residue((x-1)^(1)*SpectralCurve(Diag1Sheet1),x=1);
  EQ14:=residue((x-1)^(1)*SpectralCurve(Diag1Sheet2),x=1);
  EQ15:=residue((x-1)^(0)*SpectralCurve(Diag1Sheet1),x=1);
  EQ16:=residue((x-1)^(0)*SpectralCurve(Diag1Sheet2),x=1);
  EQ17:=residue((x-1)^(-1)*SpectralCurve(Diag1Sheet1),x=1);
  EQ18:=residue((x-1)^(-1)*SpectralCurve(Diag1Sheet2),x=1);
```

$$EQ19 := -t111\,t121 + P142 \tag{10}$$
$$EQ20 := -t111\,t121 + P142$$
$$EQ21 := 0$$
$$EQ22 := 0$$
$$EQ11 := -t110\,t121 - t111\,t120 + P132$$
$$EQ12 := -t110\,t121 - t111\,t120 + P132$$
$$EQ13 := -t010\,t111 - t020\,t111 - t110\,t120 + t111\,Unknown5 - t121\,Unknown5 + P122$$
$$EQ14 := -t010\,t121 - t020\,t121 - t110\,t120 - t111\,Unknown6 + t121\,Unknown6 + P122$$
$$EQ15 := -t010\,t110 + t010\,t111 - t020\,t110 + t020\,t111 + t110\,Unknown5 - t120\,Unknown5 + P112$$
$$EQ16 := -t010\,t120 + t010\,t121 - t020\,t120 + t020\,t121 - t110\,Unknown6 + t120\,Unknown6 + P112$$
$$EQ17 := t010\,t020 + t010\,t110 - t010\,t111 - t010\,Unknown5 + t020\,t110 - t020\,t111 - t020\,Unknown5 + Unknown5^2 + P012$$
$$EQ18 := t010\,t020 + t010\,t120 - t010\,t121 - t010\,Unknown6 + t020\,t120 - t020\,t121 - t020\,Unknown6 + Unknown6^2 + P012$$

```
> P142:=solve(EQ19,P142);
  simplify(EQ19);
  simplify(EQ20);
```

$$P142 := t111\,t121 \tag{11}$$
$$0$$

$$0$$

```
> P132:=solve(EQ11,P132);
  simplify(EQ11);
  simplify(EQ12);
  simplify(EQ13);
```

$$P132 := t110\,t121 + t111\,t120 \tag{12}$$
$$0$$
$$0$$
$$(-t010 - t020 + Unknown5)\,t111 - t110\,t120 - t121\,Unknown5 + P122$$

Summary of coefficients and additional coherence relations (sum of residues is null and P112+P012= tinfty21*tinfty10+tinfty11*tinfty20)

```
> P011:=P011;
  P111:=P111;
  P121:=P121;



  P142:=P142;
  P132:=P132;
  P122:=P122;
  P112:=P112;

  P022:=P022;
  P012:=P012;



  CoherenceEquation1;
  CoherenceEquation2;
  CoherenceEquation5;
```

$$P011 := t010 + t020 \tag{13}$$
$$P111 := t110 + t120$$
$$P121 := t111 + t121$$
$$P142 := t111\,t121$$
$$P132 := t110\,t121 + t111\,t120$$
$$P122 := P122$$
$$P112 := P112$$
$$P022 := t010\,t020$$
$$P012 := P012$$
$$tinfty10 + tinfty20 + t010 + t020 + t110 + t120$$
$$P012 + P112$$
$$t010\,t020 - tinfty10\,tinfty20 + P112 + P122$$

There are 3 unknown coefficients but two additional relations, thus only one remaining unknown coefficient